
· An SAP LUW (Logical Unit of Work) consists of changes in SAP R/3 that logically belong together. These changes are either carried out in full or they are not carried out at all ("all or nothing" principle). 

· In general, a business transaction is not processed by a single SAP LUW. The entire process from receipt of a customer order to the issue of an invoice, for example, is split into individual, logical parts. Each part corresponds to an SAP LUW. The definition of SAP LUWs depends on the entire process and its modeling.

· For further information, refer to the ABAP Editor keyword documentation for transaction processing.

· A database LUW consists of changes that are executed until the database status is "sealed" (DB Commit). 

· Within a database LUW, it is always possible to discard all the changes that have taken place up to that point (DB rollback), in which case the database is reset to the status it had before the current database LUW.  You use this DB rollback function to restore the previous (consistent) database status if an error has occurred.

· The current database status is "sealed" by a DB commit. After the database is sealed, it is no longer possible to discard the current database LUW.

· Using the ABAP statements ROLLBACK WORK and COMMIT WORK, you can explicitly implement a DB rollback or DB commit.  There are also situations where a DB commit is triggered implicitly (see next graphic).  

· For further information, refer to the ABAP Editor keyword documentation for transaction processing.

· If there is an error during processing of an SAP LUW, it should be possible to return to a consistent database status that existed before the beginning of the SAP LUW. To make this possible, the SAP LUW must be processed within a DB LUW. 

· Due to the client-server architecture of SAP R/3, this is not a trivial matter because an SAP R/3 transaction usually has several screens for entering change data. Whenever there is a screen change, the system automatically triggers an implicit DB commit (shown in the following graphics). However, it should be possible within a transaction to bundle user entries that form an SAP LUW within a DB LUW and write them to the database.

· The SAP R/3 system is based on the three-tier architecture of a client/server system. The three tiers are the database, application, and presentation servers. 

· This architecture, along with the distribution of users' requests (user dispatching), leads to a highly efficient and cost-effective multi-user system.

· The three-tier architecture means that a large number of users with low-cost desktop computers (with low performance) can be mapped to a small number of high-performance (and considerably more expensive) work processes on application servers. Each work process on an application server is assigned a work process on a high-performance database server.

· Distributing user requests to work processes means that individual clients at presentation server level are assigned to a work process for a particular period. In turn, the work process uses another work process in the database. After the work process has processed the user input in a dialog step, the user, along with the program context, is removed from the work process. This work process can then be used by another user. 

· The three-tier architecture is far more scalable than a "fat" client architecture, where the presentation and application levels run on one server. With a three-tier architecture, the number of database users is considerably lower than the number of users active in the system.  This has a positive effect on the behavior of the database.

· Before each single screen is displayed, the current work process is released on the application server. This, in turn, triggers the release of the respective database work process and automatically initiates an implicit database commit. 

· Releasing the work process before each user dialog ensures that user actions, which usually take longer, do not block any work processes on the application server or, particularly, on the database server. This reduces the load on database resources. Only when the user has completed processing the screen, which will then require internal program processing, is the user "rolled back" together with the program context. This means that the user is returned to a work process for processing. 

· Implicit DB commits are always initiated whenever the program has to wait, as in the following cases:

· When the system displays an SAP screen

· When the system sends a dialog message

· Whenever there are synchronous and asynchronous RFC calls (Remote Function Call)

· With CALL TRANSACTION <t_code>  or   SUBMIT <program> statements

· For the above-mentioned reasons, implicit DB commits or changes that belong to an SAP LUW may not be placed in different dialog steps (dialog step = program processing after a screen). The reason is these steps would thus not be within a DB LUW.

	Unit: LUW Concepts

	At the conclusion of these exercises, you will be able to:

· Assess function modules and subroutines for LUW processing suitability.



	[image: image11.wmf]
	The program SAPBC414T_BOOKINGS_01 allows you to cancel bookings for a flight. To do this, you select the respective bookings for cancellation on screen 200.

Implement the database dialog:
By clicking the Save icon (function code SAVE) on screen 200, the bookings selected in the SBOOK table are to be changed. In addition, the respective flight in table SFLIGHT must be modified within the same database LUW (booking total and number of booked places will change because of the cancellation) since the database table change for both database tables must follow the „all or nothing“ principle. Use existing function modules for the change.

	Program:

SAPMZ##_BOOKINGS1
Transaction code: 
Z##_BOOKINGS1
Template:
SAPBC414T_BOOKINGS_01
Model solution:
SAPBC414S_BOOKINGS_01


1-1
Copy the program template SAPBC414T_ BOOKINGS_01 with all subobjects to SAPMZ##_BOOKINGS1 (## is your group number) and assign transaction code Z##_BOOKINGS1 to the program.  Familiarize yourself with the program functions.

1-2
Encapsulate the ABAP statements for the database updates in the subroutine SAVE_MODIFIED_BOOKING, which is called from the PAI module USER_COMMAND_0200 (screen 200).

Perform the database update is to be performed using the available function modules. A choice of two function modules is available for each table: UPDATE_SBOOK,  UPDATE_SBOOK_A ,  UPDATE_SFLIGHT  and  UPDATE_SFLIGHT_A .
By combining the correct function modules in the correct call sequence, you ensure that, if there is an error, the data in the database tables concerned remains consistent.

1-2-1
Which function modules must be called and in what order? For this purpose, check the source code in the function modules for ABAP statements that terminate the database LUW prematurely and can, therefore, result in inconsistent data being written to the tables permanently.  

1-2-2
Call up the function modules in the appropriate order from the subroutine SAVE_MODIFIED_BOOKING.
1-2-3
Deal with the exceptions of the function modules. Possible user messages:


Flight / bookings updated
(
Message 034
Error with flight / booking update
(
Message 044
Changes not successful
(
Message 048
Flight booked out or overbooked
(
Message 045
Flight does not exist
(
Message 046

	The data records to be changed in the database table SBOOK are buffered in the internal table ITAB_SBOOK_MODIFY.

The key fields of the corresponding flight can be accessed using the WA_SFLIGHT structure.

	Use the graphic given for the exercises in the last unit to familiarize yourself with the function range of the template.

	Unit: LUW Concepts


Model Solution SAPBC414S_BOOKINGS_01
1 Module Pool
*&-----------------------------------------------------------------*

*& Modulpool         SAPBC414S_BOOKINGS_01                         *
*&-----------------------------------------------------------------*

INCLUDE BC414S_BOOKINGS_01TOP.
INCLUDE BC414S_BOOKINGS_01O01.
INCLUDE BC414S_BOOKINGS_01I01.
INCLUDE BC414S_BOOKINGS_01F01.
INCLUDE BC414S_BOOKINGS_01F02.
INCLUDE BC414S_BOOKINGS_01F03.
INCLUDE BC414S_BOOKINGS_01F04.
INCLUDE BC414S_BOOKINGS_01F05.
INCLUDE BC414S_BOOKINGS_01F06.
2 SCREEN 100
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0100.
*

PROCESS AFTER INPUT.
  MODULE EXIT AT EXIT-COMMAND.
  MODULE SAVE_OK_CODE.
  CHAIN.
* cancel booking: check if flight exists or flight can be created
    FIELD: SDYN_CONN-CARRID, SDYN_CONN-CONNID, SDYN_CONN-FLDATE.
    MODULE USER_COMMAND_0100.
  ENDCHAIN.
SCREEN 200
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0200.
  MODULE TRANS_DETAILS.
  CALL SUBSCREEN SUB1 INCLUDING SY-CPROG '0201'.
  LOOP AT ITAB_BOOK INTO WA_BOOK WITH CONTROL TC_SBOOK.
    MODULE TRANS_TO_TC.
* allow only modification of bookings, that are not allready cancelled
    MODULE MODIFY_SCREEN.
  ENDLOOP.
*

PROCESS AFTER INPUT.
  LOOP AT ITAB_BOOK.
* mark changed bookings in internal table itab_book
    FIELD SDYN_BOOK-CANCELLED MODULE MODIFY_ITAB ON REQUEST.
  ENDLOOP.
  MODULE EXIT AT EXIT-COMMAND.
  MODULE SAVE_OK_CODE.
  MODULE USER_COMMAND_0200.
3 SCREEN 201
PROCESS BEFORE OUTPUT.
PROCESS AFTER INPUT.
4 SCREEN 300
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0300.
  MODULE TABSTRIP_INIT.
  MODULE TRANS_DETAILS.
  CALL SUBSCREEN TAB_SUB INCLUDING SY-CPROG SCREEN_NO.
*

PROCESS AFTER INPUT.
  CALL SUBSCREEN TAB_SUB.
  MODULE EXIT AT EXIT-COMMAND.
  MODULE SAVE_OK_CODE.
  MODULE TRANS_FROM_0300.
  MODULE USER_COMMAND_0300.
5 SCREEN 301
PROCESS BEFORE OUTPUT.
* MODULE HIDE_BOOKID.
PROCESS AFTER INPUT.
6 SCREEN 302
PROCESS BEFORE OUTPUT.
PROCESS AFTER INPUT.
7 SCREEN 303
PROCESS BEFORE OUTPUT.
PROCESS AFTER INPUT.
8 TOP Include
*&-----------------------------------------------------------------*

*& Include BC414S_BOOKINGS_01TOP                                   *
*&-----------------------------------------------------------------*

PROGRAM  sapbc414s_bookings_01 MESSAGE-ID bc414.
* line type of internal table itab_book, used to display bookings in
* table control
TYPES: BEGIN OF wa_book_type.
INCLUDE: STRUCTURE sbook.
TYPES:   name TYPE scustom-name,
         mark,
       END OF wa_book_type.
* work area and internal table used to display bookings in table 
* control
DATA: wa_book TYPE wa_book_type,
      itab_book TYPE TABLE OF wa_book_type.
* bookings to be modified on database table
DATA: itab_sbook_modify TYPE TABLE OF sbook.
* change documents: bookings before changes are performed
DATA: itab_cd TYPE TABLE OF sbook WITH NON-UNIQUE KEY
      carrid connid fldate bookid customid.
* work areas for database tables spfli, sflight, sbook.
DATA: wa_sbook TYPE sbook, wa_sflight TYPE sflight, wa_spfli TYPE 
      spfli.
* complex transactions: number of the customer created in the called
* transaction
data: scust_id(20).
* transport function codes from screens
DATA: ok_code TYPE sy-ucomm, save_ok LIKE ok_code.
* define subscreen screen number on tabstrip, screen 300
DATA: screen_no TYPE sy-dynnr.
* used to handle sy-subrc, which is determined in form
DATA  sysubrc LIKE sy-subrc.
* transporting fields to/from screen
TABLES: sdyn_conn, sdyn_book.
* table control declaration (display bookings),
* tabstrip declaration (create booking)
CONTROLS: tc_sbook TYPE TABLEVIEW USING SCREEN '0200',
          tab TYPE TABSTRIP.
9 PBO Modules
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_01O01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0100 OUTPUT.
  SET PF-STATUS 'DYN_100'.
  SET TITLEBAR 'DYN_100'.
ENDMODULE.                             " STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  STATUS_0200  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0200 OUTPUT.
  SET PF-STATUS 'DYN_200'.
  SET TITLEBAR 'DYN_200' WITH sdyn_conn-carrid sdyn_conn-connid
                              sdyn_conn-fldate.
ENDMODULE.                             " STATUS_0200  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  STATUS_0300  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0300 OUTPUT.
  SET PF-STATUS 'DYN_300'.
  SET TITLEBAR 'DYN_300' WITH sdyn_conn-carrid sdyn_conn-connid
                              sdyn_conn-fldate.
ENDMODULE.                             " STATUS_0300  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_DETAILS  OUTPUT
*&-----------------------------------------------------------------*

MODULE trans_details OUTPUT.
  MOVE-CORRESPONDING: wa_spfli   TO sdyn_conn,
                      wa_sflight TO sdyn_conn,
                      wa_sbook   TO sdyn_book.
ENDMODULE.                             " TRANS_DETAILS  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_TO_TC  OUTPUT
*&-----------------------------------------------------------------*

MODULE trans_to_tc OUTPUT.
  MOVE-CORRESPONDING wa_book TO sdyn_book.
ENDMODULE.                             " TRANS_TO_TC  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  MODIFY_SCREEN  OUTPUT
*&-----------------------------------------------------------------*

MODULE modify_screen OUTPUT.
  LOOP AT SCREEN.
    CHECK screen-name = 'SDYN_BOOK-CANCELLED'.
    CHECK ( NOT sdyn_book-cancelled IS INITIAL ) AND
          ( sdyn_book-mark IS INITIAL ).
    screen-input = 0.
    MODIFY SCREEN.
  ENDLOOP.
ENDMODULE.                             " MODIFY_SCREEN  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TABSTRIP_INIT  OUTPUT
*&-----------------------------------------------------------------*

MODULE tabstrip_init OUTPUT.
  CHECK tab-activetab IS INITIAL.
  tab-activetab = 'BOOK'.
  screen_no = '0301'.
ENDMODULE.                             " TABSTRIP_INIT  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  HIDE_BOOKID  OUTPUT
*&-----------------------------------------------------------------*

MODULE hide_bookid OUTPUT.
* hide field displaying customer number when working with number range
* object BS_SCUSTOM
  LOOP AT SCREEN.
    CHECK screen-name = 'SDYN_BOOK-BOOKID'.
    screen-active = 0.
    MODIFY SCREEN.
  ENDLOOP.
ENDMODULE.                             " HIDE_BOOKID  OUTPUT
10 PAI Modules
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_01I01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'CANCEL'.
      CASE sy-dynnr.
        WHEN '0100'.
          LEAVE PROGRAM.
        WHEN '0200'.
          LEAVE TO SCREEN '0100'.
        WHEN '0300'.
          LEAVE TO SCREEN '0100'.
        WHEN OTHERS.
      ENDCASE.
    WHEN 'EXIT'.
      LEAVE PROGRAM.
    WHEN OTHERS.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

MODULE save_ok_code INPUT.
  save_ok = ok_code.
  CLEAR ok_code.
ENDMODULE.                             " SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
****************************CANCEL BOOKING**************************
    WHEN 'BOOKC'.
      PERFORM read_sflight USING wa_sflight sysubrc.
* process returncode - if flight does not exist: e-message
      PERFORM process_sysubrc_bookc.
      PERFORM read_spfli USING wa_spfli.
      PERFORM read_sbook USING itab_book itab_cd.
      REFRESH CONTROL 'TC_SBOOK' FROM SCREEN '0200'.
****************************CREATE BOOKING**************************
    WHEN 'BOOKN'.
      PERFORM read_sflight USING wa_sflight sysubrc.
* process returncode - if flight does not exist: e-message
      PERFORM process_sysubrc_bookn.
      PERFORM read_spfli USING wa_spfli.
      PERFORM initialize_sbook USING wa_sbook.
    WHEN 'BACK'.
      SET SCREEN 0.
    WHEN OTHERS.
      SET SCREEN '0100'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0200 INPUT.
  CASE save_ok.
    WHEN 'SAVE'.
* collect marked (changed) data sets in seperate internal table
      PERFORM collect_modified_data USING itab_sbook_modify.
* perform database changes
      PERFORM save_modified_booking.
      SET SCREEN '0100'.
    WHEN 'BACK'.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0200'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

*&      Module  MODIFY_ITAB  INPUT
*&-----------------------------------------------------------------*

MODULE modify_itab INPUT.
  wa_book-cancelled = sdyn_book-cancelled.
  wa_book-mark = 'X'.
  MODIFY itab_book FROM wa_book INDEX tc_sbook-current_line.
ENDMODULE.                             " MODIFY_ITAB  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0300  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0300 INPUT.
  PERFORM tabstrip_set.
  CASE save_ok.
    WHEN 'NEW_CUSTOM'.
      PERFORM create_new_customer.
      SET SCREEN '0300'.
    WHEN 'SAVE'.
      PERFORM save_new_booking.
      SET SCREEN '0100'.
    WHEN 'BACK'.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0300'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0300  INPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_FROM_0300  INPUT
*&-----------------------------------------------------------------*

MODULE trans_from_0300 INPUT.
  MOVE-CORRESPONDING sdyn_book TO wa_sbook.
ENDMODULE.                             " TRANS_FROM_0300  INPUT
11 FORM Routines
12 F01
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_01F01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  COLLECT_MODIFIED_DATA
*&-----------------------------------------------------------------*

*      -->P_ITAB_SBOOK_MODIFY  text
*------------------------------------------------------------------*

FORM collect_modified_data USING p_itab_sbook_modify
                                 LIKE itab_sbook_modify.
  DATA: wa_book LIKE LINE OF itab_book,
        wa_sbook_modify LIKE LINE OF p_itab_sbook_modify.
  CLEAR: p_itab_sbook_modify.
* Only bookings are collected, that
* 1) have been changed (mark = 'X')
* 2) shall be cancelled (cancelled = 'X')
  LOOP AT itab_book INTO wa_book
       WHERE    mark = 'X'
       AND cancelled = 'X'.
    MOVE-CORRESPONDING wa_book TO wa_sbook_modify.
    APPEND wa_sbook_modify TO p_itab_sbook_modify.
  ENDLOOP.
ENDFORM.                               " COLLECT_MODIFIED_DATA
*&-----------------------------------------------------------------*

*&      Form  INITIALIZE_SBOOK
*&-----------------------------------------------------------------*

*      -->P_WA_SBOOK  text
*------------------------------------------------------------------*

FORM initialize_sbook USING p_wa_sbook TYPE sbook.
  CLEAR p_wa_sbook.
  MOVE-CORRESPONDING wa_sflight TO p_wa_sbook.
ENDFORM.                               " INITIALIZE_SBOOK
*&-----------------------------------------------------------------*

*&      Form  PROCESS_SYSUBRC_BOOKC
*&-----------------------------------------------------------------*

FORM process_sysubrc_bookc.
  CASE sysubrc.
    WHEN 0.
      SET SCREEN '0200'.
    WHEN OTHERS.
      MESSAGE e023 WITH sdyn_conn-carrid sdyn_conn-connid
                        sdyn_conn-fldate.
  ENDCASE.
ENDFORM.                               " PROCESS_SYSUBRC_BOOKC
*&-----------------------------------------------------------------*

*&      Form  PROCESS_SYSUBRC_BOOKN
*&-----------------------------------------------------------------*

FORM process_sysubrc_bookn.
  CASE sysubrc.
    WHEN 0.
      SET SCREEN '0300'.
    WHEN OTHERS.
      MESSAGE e023 WITH sdyn_conn-carrid sdyn_conn-connid
                        sdyn_conn-fldate.
  ENDCASE.
ENDFORM.                               " PROCESS_SYSUBRC_BOOKN
*&-----------------------------------------------------------------*

*&      Form  TABSTRIP_SET
*&-----------------------------------------------------------------*

FORM tabstrip_set.
  IF save_ok = 'BOOK' OR save_ok = 'DETCON' OR save_ok = 'DETFLT'.
    tab-activetab = save_ok.
  ENDIF.
  CASE save_ok.
    WHEN 'BOOK'.
      screen_no = '0301'.
    WHEN 'DETCON'.
      screen_no = '0302'.
    WHEN 'DETFLT'.
      screen_no = '0303'.
  ENDCASE.
ENDFORM.                               " TABSTRIP_SET
*&-----------------------------------------------------------------*

*&      Form  CREATE_NEW_CUSTOMER
*&-----------------------------------------------------------------*

FORM create_new_customer.
********************** TO BE IMPLEMENTED LATER *********************
ENDFORM.                               " CREATE_NEW_CUSTOMER
*&-----------------------------------------------------------------*

*&      Form  NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

FORM number_get_next USING p_wa_sbook LIKE sbook.
********************** TO BE IMPLEMENTED LATER *********************
ENDFORM.                               " NUMBER_GET_NEXT
13 F02
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_01F02
*------------------------------------------------------------------*

*------------------------------------------------------------------*

*   FORM ENQ_SFLIGHT
*------------------------------------------------------------------*

FORM enq_sflight.
********************* TO BE IMPLEMENTED LATER **********************
ENDFORM.                               "ENQ_SFLIGHT
*------------------------------------------------------------------*

*   FORM ENQ_SBOOK
*------------------------------------------------------------------*

FORM enq_sbook.
********************* TO BE IMPLEMENTED LATER **********************
ENDFORM.                               "ENQ_SBOOK
*------------------------------------------------------------------*

*   FORM ENQ_SFLIGHT_SBOOK
*------------------------------------------------------------------*

FORM enq_sflight_sbook.
********************* TO BE IMPLEMENTED LATER **********************
ENDFORM.                               "ENQ_SFLIGHT_SBOOK
*------------------------------------------------------------------*

*   FORM DEQ_ALL
*------------------------------------------------------------------*

FORM deq_all.
********************* TO BE IMPLEMENTED LATER **********************
ENDFORM.                               "DEQ_ALL
14 F03
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_01F03
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  READ_SFLIGHT
*&-----------------------------------------------------------------*

*      -->P_WA_SFLIGHT  text
*      -->P_SYSUBRC     text
*------------------------------------------------------------------*

FORM read_sflight USING p_wa_sflight TYPE sflight
                        p_sysubrc LIKE sy-subrc.
  SELECT SINGLE * FROM sflight INTO p_wa_sflight
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid
         AND   fldate = sdyn_conn-fldate.
  p_sysubrc = sy-subrc.
ENDFORM.                               " READ_SFLIGHT
*&-----------------------------------------------------------------*

*&      Form  READ_SBOOK
*&-----------------------------------------------------------------*

*      -->P_ITAB_BOOK  text
*      -->P_ITAB_CD    text
*------------------------------------------------------------------*

FORM read_sbook USING p_itab_book LIKE itab_book
                      p_itab_cd   LIKE itab_cd.
  TYPES: BEGIN OF wa_custom_type,
           id TYPE scustom-id,
           name TYPE scustom-name,
         END OF wa_custom_type.
  DATA: wa_custom TYPE wa_custom_type,
        itab_custom TYPE STANDARD TABLE OF wa_custom_type
        WITH NON-UNIQUE KEY id,
        wa_book LIKE LINE OF p_itab_book,
        wa_cd   LIKE LINE OF p_itab_cd.
  CLEAR: p_itab_book, p_itab_cd.
* Select customer names in buffer table (array fetch)
  SELECT id name FROM scustom INTO CORRESPONDING FIELDS
         OF TABLE itab_custom.
* Select all bookings on selected flight (array fetch)
  SELECT * FROM sbook INTO CORRESPONDING FIELDS OF TABLE p_itab_book
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid
         AND   fldate = sdyn_conn-fldate.
* read customer names corresponding to customer number from buffer 
* table
  LOOP AT p_itab_book INTO wa_book.
    READ TABLE itab_custom INTO wa_custom WITH TABLE KEY
                                            id = wa_book-customid.
    wa_book-name = wa_custom-name.
    MODIFY p_itab_book FROM wa_book.
    MOVE-CORRESPONDING wa_book TO wa_cd.
    APPEND wa_cd TO p_itab_cd.
  ENDLOOP.
  SORT p_itab_book BY bookid customid.
ENDFORM.                               " READ_SBOOK
*&-----------------------------------------------------------------*

*&      Form  READ_SPFLI
*&-----------------------------------------------------------------*

*      -->P_WA_SPFLI  text
*------------------------------------------------------------------*

FORM read_spfli USING p_wa_spfli TYPE spfli.
  SELECT SINGLE * FROM spfli INTO p_wa_spfli
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid.
  IF sy-subrc <> 0.
    MESSAGE e022 WITH sdyn_conn-carrid sdyn_conn-connid.
  ENDIF.
ENDFORM.                               " READ_SPFLI
15 F04
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_01F04
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  SAVE_MODIFIED_BOOKING
*&-----------------------------------------------------------------*

FORM save_modified_booking.
* Modify data on database tables sbook and sflight
  CALL FUNCTION 'UPDATE_SBOOK'
       EXPORTING
            itab_sbook     = itab_sbook_modify
       EXCEPTIONS
            update_failure = 1
            OTHERS         = 2.
  CASE sy-subrc.
    WHEN 0.
      PERFORM update_sflight.
    WHEN OTHERS.
      MESSAGE a044 WITH wa_sflight-carrid wa_sflight-connid
                        wa_sflight-fldate.
  ENDCASE.
ENDFORM.                               " SAVE_MODIFIED_BOOKING
*&-----------------------------------------------------------------*

*&      Form  UPDATE_SFLIGHT
*&-----------------------------------------------------------------*

FORM update_sflight.
  CALL FUNCTION 'UPDATE_SFLIGHT'
       EXPORTING
            carrier          = wa_sflight-carrid
            connection       = wa_sflight-connid
            date             = wa_sflight-fldate
       EXCEPTIONS
            update_failure   = 1
            flight_full      = 2
            flight_not_found = 3
            OTHERS           = 4.
  CASE sy-subrc.
    WHEN 0.
      MESSAGE s034 WITH wa_sflight-carrid wa_sflight-connid
                        wa_sflight-fldate.
    WHEN 1.
      MESSAGE a044 WITH wa_sflight-carrid wa_sflight-connid
                        wa_sflight-fldate.
    WHEN 2.
      MESSAGE a045.
    WHEN 3.
      MESSAGE a046.
    WHEN OTHERS.
      MESSAGE a048.
  ENDCASE.
ENDFORM.                               " UPDATE_SFLIGHT
*&-----------------------------------------------------------------*

*&      Form  SAVE_NEW_BOOKING
*&-----------------------------------------------------------------*

FORM save_new_booking.
********************* TO BE IMPLEMENTED LATER **********************
ENDFORM.                               " SAVE_NEW_BOOKING
16 F05
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_01F05
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  CONVERT_TO_LOC_CURRENCY
*&-----------------------------------------------------------------*

*      -->P_WA_SBOOK  text
*------------------------------------------------------------------*

FORM convert_to_loc_currency USING p_wa_sbook TYPE sbook.
  SELECT SINGLE currcode FROM scarr INTO p_wa_sbook-loccurkey
         WHERE carrid = p_wa_sbook-carrid.
  CALL FUNCTION 'CONVERT_TO_LOCAL_CURRENCY_N'
       EXPORTING
            client           = sy-mandt
            date             = sy-datum
            foreign_amount   = p_wa_sbook-forcuram
            foreign_currency = p_wa_sbook-forcurkey
            local_currency   = p_wa_sbook-loccurkey
       IMPORTING
            local_amount     = p_wa_sbook-loccuram
       EXCEPTIONS
            no_rate_found    = 1
            overflow         = 2
            no_factors_found = 3
            no_spread_found  = 4
            derived_2_times  = 5
            OTHERS           = 6.
  IF sy-subrc <> 0.
    MESSAGE e080 WITH sy-subrc.
  ENDIF.
ENDFORM.                               " CONVERT_TO_LOC_CURRENCY
17 F06
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_01F06
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  CREATE_CHANGE_DOCUMENTS
*&-----------------------------------------------------------------*

FORM create_change_documents.
********************* TO BE IMPLEMENTED LATER **********************
ENDFORM.                               " CREATE_CHANGE_DOCUMENTS

